Supplementary files

An integral pharmacokinetic analysis of piperacillin and tazobactam in plasma and urine in critically ill patients

Authors

Eveline Wallenburg[1], Rob ter Heine[1], Jeroen A. Schouten[2,3], Jelmer Raaijmakers[4], Jaap ten Oever[3,5], Eva Kolwijck[6], David M. Burger[1,3], Peter Pickkers[2,3], Tim Frenzel[2,3], Roger J.M. Brüggemann[1,3]

Affiliation

1 Radboud university medical center, Radboud Institute for Health Sciences, Department of Pharmacy, Nijmegen, The Netherlands

2 Department of Intensive Care, Radboud university medical center, Nijmegen, The Netherlands

3. Radboud Center for Infectious diseases (RCI), Radboud university medical center, Nijmegen, The Netherlands

4. Department of Medical Microbiology, Radboud university medical center, Nijmegen, The Netherlands

5 Department of Internal Medicine, Radboud university medical center, Nijmegen, The Netherlands

6. Department of Medical Microbiology, Jeroen Bosch Ziekenhuis, Den Bosch, The Netherlands

Corresponding author:

Eveline Wallenburg

Department of Pharmacy 864, Radboudumc, Geert Grooteplein Zuid 10, 6525 GA Nijmegen, The Netherlands

eveline.wallenburg@radboudumc.nl

Content

Goodness-of-fit plots

Control stream NONMEM

Goodness-of-fit plots



Figure S1: Goodness-of-fit diagnostics for the final pharmacokinetic model of piperacillin in the central compartment. The population and individual predicted concentrations are in concordance with the observed concentrations; the discrepancy between predictions and observations is small. Furthermore, the conditional weighted residuals indicate no model misspecification, the distribution is homogeneous and the majority of the data lie within the (−2 to +2) interval.



Figure S2: Goodness-of-fit diagnostics for the final pharmacokinetic model of tazobactam in the central compartment. The population and individual predicted concentrations are in concordance with the observed concentrations; the discrepancy between predictions and observations is small. Furthermore, the conditional weighted residuals indicate no model misspecification, the distribution is homogeneous and the majority of the data lie within the (−2 to +2) interval.



Figure S3: Goodness-of-fit diagnostics for the final pharmacokinetic model of piperacillin in the urine output compartment. The population and individual predicted concentrations are in concordance with the observed concentrations; the discrepancy between predictions and observations is small. Furthermore, the conditional weighted residuals indicate no model misspecification, the distribution is homogeneous and the majority of the data lie within the (−2 to +2) interval.



Figure S4: Goodness-of-fit diagnostics for the final pharmacokinetic model of tazobactam in the urine output compartment. The population and individual predicted concentrations are in concordance with the observed concentrations; the discrepancy between predictions and observations is small. Furthermore, the conditional weighted residuals indicate no model misspecification, the distribution is homogeneous and the majority of the data lie within the (−2 to +2) interval.



Figure S5: Visual predictive check (VPC) for the final pharmacokinetic model stratified for compartment: central piperacillin compartment (CMT=1) central tazobactam compartment (CMT=3), urine piperacillin compartment (CMT=5) and urine tazobactam compartment (CMT=6). The distribution of observed concentrations was consistent with the distribution of the predicted concentrations, suggesting a good internal validity of the model.

Control stream NONMEM

```
$PROBLEM PK

$INPUT ID TIME AMT RATE DV UVOL MDV CMT EVID OCC AGE SEX WT HT MDRD CKDCR CKDCYS CKDCRCYS
CREATCL

$DATA piptazo.csv  IGNORE=@

$SUBROUTINES ADVAN13 TRANS=1 TOL=9

$MODEL

COMP=(CENTRPIP)
COMP=(PERIPIP)
COMP=(CENTRTAZ)
COMP=(PERITAZ)
COMP=(URINEPIP)
COMP=(URINETAZ)

$PK
; --- TIME AFTER DOSE CALCULATION

IF (AMT.GT.0) THEN
TDOS=TIME
TAD=0.0
ENDIF
IF (AMT.EQ.0) TAD=TIME-TDOS

; --- FAT-FREE MASS CALCULATION

BMI=WT/(HT/100)**2

IF (SEX.EQ.1) FFM=(9270*WT)/(6680+(216*BMI)) ; FAT-FREE MASS IN MALES
IF (SEX.EQ.0) FFM=(9270*WT)/(8780+(244*BMI)) ; FAT-FREE MASS IN FEMALES

; --- ALLOMETRY SCALED TO 1.80 MALE OF 70 KG
ALLOCL=(FFM/58.2)**0.75
ALLOV=(FFM/58.2)

; --- OCCASSION
IF (OCC.EQ.1) IOV=ETA(4)
IF (OCC.EQ.2) IOV=ETA(5)

; --- PIPERACILLIN PK

TVCLPIP=THETA(1)*ALLOCL
CLPIP=TVCLPIP*EXP(ETA(1))

TVCLRPIP=THETA(2)*CREATCL
```

```
CLRPIP=TVCLRPIP*EXP(IOV)

TVV1PIP=THETA(3)*ALLOV
V1PIP=TVV1PIP*EXP(ETA(2))

TVV2PIP=THETA(4)*ALLOV
V2PIP=TVV2PIP*EXP(ETA(3))

TVQPIP=THETA(5)*ALLOCL
QPIP=TVQPIP

;--- TAZOBACTAM PK

TVCLTAZ=THETA(6)*ALLOCL
CLTAZ=TVCLTAZ*EXP(ETA(1))

CLRTAZ=CLRPIP*THETA(7)

TVV1TAZ=THETA(8)*ALLOV
V1TAZ=TVV1TAZ*EXP(ETA(2))

TVV2TAZ=THETA(9)*ALLOV
V2TAZ=TVV2TAZ*EXP(ETA(3))

TVQTAZ=THETA(10)*ALLOCL
QTAZ=TVQTAZ

S1=V1PIP
S3=V1TAZ
S5=UVOL
S6=UVOL

K10=CLPIP/V1PIP
K12=QPIP/V1PIP
K21=QPIP/V2PIP
K15=CLRPIP/V1PIP

K30=CLTAZ/V1TAZ
K34=QTAZ/V1TAZ
K43=QTAZ/V2TAZ
K36=CLRTAZ/V1TAZ

$DES
DADT(1)=-K12*A(1)+K21*A(2)-K10*A(1)-K15*A(1)
DADT(2)=K12*A(1)-K21*A(2)
DADT(3)=-K34*A(3)+K43*A(4)-K30*A(3)-K36*A(3)
DADT(4)=K34*A(3)-K43*A(4)
DADT(5)=K15*A(1)
DADT(6)=K36*A(3)
```

```
$THETA
4.11      ; CL NONRENAL PIPERACILLIN
0.0613    ; CL RENAL PIPERACILLIN (GRADIENT)
29.7      ; V1 PIPERACILLIN
4.26      ; V2 PIPERACILLIN
3.27      ; Q PIPERACILLIN
2.49      ; CL NONRENAL TAZOBACTAM
0.0674    ; CL RENAL TAZOBACTAM (GRADIENT)
30.3      ; V1 TAZOBACTAM
4.56      ; V2 TAZOBACTAM
3.44      ; Q TAZOBACTAM

$OMEGA
0.651     ; IIV CL NONRENAL PIPERACILLIN/TAZOBACTAM
0.177     ; IIV V1 PIPERACILLIN/TAZOBACTAM
2.51      ; IIV V2 PIPERACILLIN/TAZOBACTAM

$OMEGA BLOCK(1) 0.0946 ; IOV
$OMEGA BLOCK(1) SAME   ; IOV

$ERROR
CCP=A(1)/V1PIP
CCT=A(3)/V1TAZ
UVL=UVOL
IF(UVL<=0.0) UVL=1.0
CUP=A(5)/UVL
CUT=A(6)/UVL

IF (MDV.EQ.0) DUMMY=0
IF (MDV.EQ.1) DUMMY=0.0000000001

IF(CMT.EQ.1) THEN
IPRED=CCP
Y=LOG(IPRED+DUMMY)+ERR(1)
ENDIF

IF(CMT.EQ.3) THEN
IPRED=CCT
Y=LOG(IPRED+DUMMY)+ERR(2)
ENDIF

IF(CMT.EQ.5) THEN
IPRED=CUP
Y=LOG(IPRED+DUMMY)+ERR(3)
ENDIF

IF(CMT.EQ.6) THEN
IPRED=CUT
Y=LOG(IPRED+DUMMY)+ERR(4)
```

```
ENDIF

LOGIPRED=0
IF (IPRED.NE.0) LOGIPRED=LOG(IPRED)

$SIGMA
0.0755   ; PROP ERR CENTRAL PIPERACILLIN
0.0626   ; PROP ERR CENTRAL TAZOBACTAM
0.00416 ; PROP ERR URINE PIPERACILLIN
0.0199   ; PROP ERR URINE TAZOBACTAM

$EST METHOD=1 MAXEVAL=3000 SIG=3 PRINT=5 NOHABORT
$COV PRINT=E MATRIX=S UNCONDITIONAL
```